* Encoding: windows-1252.

* DELIBERATIVE QUALITIES OF NEWS FRAMES -- Analytical Code

* Code by: Eike Mark Rinke (University of Mannheim)
* This Version: 31 January 2012

*Note: Make sure to change working directory to folder containing	data before running code.

GET 
  FILE='PC_2013_GenFram_Data.sav'.

* Log-linear analysis, step 1: Hierarchical log-linear analysis, all effects (item level) .
HILOGLINEAR it_sg_bin(0 1) it_c_bin(0 1) it_incivil_bin(0 1) it_reas_bin(0 1) it_periphery_bin(0 1) 
  /METHOD=BACKWARD 
  /CRITERIA MAXSTEPS(30) P(.05) ITERATION(20) DELTA(.5) 
  /PRINT=FREQ RESID ASSOCIATION ESTIM 
  /DESIGN.

* Log-linear analysis, step 2: Test model fit without last three-way interaction retained in hierarchical analysis (item level) .
HILOGLINEAR it_sg_bin(0 1) it_c_bin(0 1) it_incivil_bin(0 1) it_reas_bin(0 1) it_periphery_bin(0 1) 
  /CRITERIA ITERATION(20) DELTA(.5) 
  /PRINT=FREQ RESID ASSOCIATION ESTIM 
  /DESIGN it_c_bin*it_incivil_bin it_c_bin*it_reas_bin it_sg_bin*it_incivil_bin it_sg_bin*it_reas_bin it_sg_bin*it_periphery_bin it_incivil_bin*it_reas_bin it_incivil_bin*it_periphery_bin.

* Univariate distributions of strategic game and contestation frames .
FREQUENCIES VARIABLES=it_sg_bin it_c_bin 
  /BARCHART PERCENT 
  /ORDER=VARIABLE .

* Strategic game hypotheses: H1, H2, and H3 (plus definitional assumption that strategic game associated with lower inclusiveness) .
CROSSTABS 
  /TABLES=it_periphery_bin it_incivil_bin it_c_bin it_reas_bin BY it_sg_bin 
  /FORMAT=AVALUE TABLES 
  /STATISTICS=CHISQ 
  /CELLS=COUNT COLUMN 
  /COUNT ROUND CELL.
EXECUTE .

* Contestation hypotheses: H4, H5, and H6 .
CROSSTABS 
  /TABLES=it_periphery_bin it_incivil_bin it_reas_bin BY it_c_bin 
  /FORMAT=AVALUE TABLES 
  /STATISTICS=CHISQ 
  /CELLS=COUNT COLUMN 
  /COUNT ROUND CELL.
EXECUTE .

* Endnote 6: Overlap of strategic game and contestation frames .
CROSSTABS 
  /TABLES=it_c_bin BY it_sg_bin 
  /FORMAT=AVALUE TABLES 
  /STATISTICS=CHISQ 
  /CELLS=COUNT COLUMN ROW 
  /COUNT ROUND CELL.
EXECUTE .

      *** Control/validation analysis (not reported in manuscript***
* Control loglinear analysis on utterance level, with all variable as binaries.
HILOGLINEAR utt_sg_bin(0 1) utt_c_bin(0 1) utt_incivil_bin(0 1) utt_reas_bin(0 1) utt_speaktype(1 3) 
  /METHOD=BACKWARD 
  /CRITERIA MAXSTEPS(30) P(.05) ITERATION(20) DELTA(.5) 
  /PRINT=FREQ RESID ASSOCIATION ESTIM 
  /DESIGN.
EXECUTE .
